Bessel functions

1 Derivation with the Frobenius method

Solution of the differential equation by the Frobenius method, but for z << 1 (not enough to derive the full solution)

```
→ besselexpand: true;
```

besselexpand true

```
→ declare(s,constant);
```

(%o2) done

We expand up to 6th order but because there are derivatives up to the 2nd order, we shouldn't use y(z) more than up to 4th order

```
→ y(z):=sum(a[n]·z^(n+s),n,0,inf);
```

(%o3) $$y(z):=\sum_{n=0}^{\infty}\left(a_n\, z^{n+s}\right)$$

Bessel differential equation

```
→ eq1: z^2·diff(y(z),z,2) + z·diff(y(z),z) + (z^2−v^2)·y(z) = 0;
```

eq1 $$(z^2-v^2)\left(\sum_{n=0}^{\infty}\left(a_n\, z^{n+s}\right)\right)+z\left(\sum_{n=0}^{\infty}\left((n+s)\, a_n\, z^{n+s-1}\right)\right)+z^2\sum_{n=0}^{\infty}\left((n+s-1)\,(n+s)\, a_n\, z^{n+s-2}\right)=0$$

```
→ eq2: intosum(eq1);
```

eq2 $$\left(\sum_{n=0}^{\infty}\left((n+s-1)\,(n+s)\, a_n\, z^{n+s}\right)\right)+\left(\sum_{n=0}^{\infty}\left((n+s)\, a_n\, z^{n+s}\right)\right)+\sum_{n=0}^{\infty}\left(a_n\, z^{n+s}\,(z^2-v^2)\right)=0$$

```
→ sum3: expand(part(eq2,1,3));
```

sum3 $$\sum_{n=0}^{\infty}\left(a_n\, z^{n+s+2}-a_n\, v^2\, z^{n+s}\right)$$

```
→ sum4: sum(part(sum3,1,1),n,0,inf);
```

sum4 $$\sum_{n=0}^{\infty}\left(a_n\, z^{n+s+2}\right)$$

```
→ sum5: intosum(sum(part(sum3,1,2),n,0,inf));
```

sum5 $$\sum_{n=0}^{\infty}\left(-\left(a_n\, v^2\, z^{n+s}\right)\right)$$

```
→ sum4a: changevar(sum4,n+2−m,m,n);
```

sum4a $$\sum_{m=2}^{\infty}\left(a_{m-2}\, z^{m+s}\right)$$

```
→ eq3: intosum(substpart(sum4a+sum5,eq2,1,3));
```

eq3 $$\left(\sum_{n=0}^{\infty}\left(-\left(a_n\, v^2\, z^{n+s}\right)\right)\right)+\left(\sum_{n=0}^{\infty}\left((n+s-1)\,(n+s)\, a_n\, z^{n+s}\right)\right)+\left(\sum_{n=0}^{\infty}\left((n+s)\, a_n\, z^{n+s}\right)\right)+\sum_{m=2}^{\infty}\left(a_{m-2}\, z^{m+s}\right)=0$$

→ eq4: ratsimp(sumcontract(eq3));

eq4 $$\left(\sum_{n=2}^{\infty}((-(a_n v^2)+(n^2+2\,s\,n+s^2)\,a_n+a_{n-2})\,z^{n+s})\right)+z^s\left(((s^2+2\,s+1)\,a_1-a_1\,v^2)\,z-a_0\,v^2+s^2\,a_0\right)=0$$

→ eqn0: -a[0]·v^2+s^2·a[0]= 0;

eqn0 $$s^2\,a_0-a_0\,v^2=0$$

If a[0] != 0, then s = +/- v. We take s = v

→ eq5: ratsubst(v,s,eq4);

eq5 $$\left(\sum_{n=2}^{\infty}((2\,n\,a_n\,v+n^2\,a_n+a_{n-2})\,z^{v+n})\right)+(2\,a_1\,v+a_1)\,z^{v+1}=0$$

→ eqn1: part(eq5,1,2,1) = 0;

eqn1 $$2\,a_1\,v+a_1=0$$

→ solve(eqn1,a[1]);

(%o15) $$[a_1=0]$$

→ eqnn: part(eq5,1,1,1,1) = 0;

eqnn $$2\,n\,a_n\,v+n^2\,a_n+a_{n-2}=0$$

→ solnn: solve(eqnn,a[n])[1];

solnn $$a_n=-\left(\frac{a_{n-2}}{2\,n\,v+n^2}\right)$$

→ /* if n = 3, we need a[3-2] = a[1], but this is zero, so a[3] is zero as well. So, all the even n are zero */
solnn1: (ratsubst(2·m,n,solnn));

solnn1 $$a_{2\,m}=-\left(\frac{a_{2\,m-2}}{4\,m\,v+4\,m^2}\right)$$

if m = 2, i.e., for a[2*2] = a[4], we need a[4-2] = a[2], which is in terms of a[2-2]=a[0].

For a[6], we need a[4], which is itself in terms of a[0], so a[6] also contains a[0].

If we continue, we see that products of coefficients are formed, that can be written in terms of factorials. Choosing a normalization

a[0] = 1/(2^v*v!)

we arrive at the standard definition of the Bessel functions J_v(z).

To ilustrate, let's take the expansion for z << 1, up to 6th order. So:

→ /* n = 2 */
a2: ratsubst(2,n,solnn);

a2 $$a_2=-\left(\frac{a_0}{4\,v+4}\right)$$

→ /* n = 4 */
a4: ratsubst(4,n,solnn);

a4 $$a_4=-\left(\frac{a_2}{8\,v+16}\right)$$

→ a4a: factor(ratsubst(rhs(a2),a[2],a4));

a4a $$a_4=\frac{a_0}{32\,(v+1)\,(v+2)}$$

→ /* n = 6 */
a6: ratsubst(6,n,solnn);

a6 $$a_6=-\left(\frac{a_4}{12\,v+36}\right)$$

→ a6a: factor(ratsubst(rhs(a4a),a[4],a6));

a6a $a_6 = -\left(\frac{a_0}{384\ (v+1)\ (v+2)\ (v+3)}\right)$

```
/* Together, we have for J_v(z) for z << 1 */
besselj_artisanal: a[0]·z^v + rhs(a2)·z^(v+2) + rhs(a4a)·z^(v+4) + rhs(a6a)·z^(v+6);
```

besselj_artisanal $-\left(\frac{a_0\ z^{v+6}}{384\ (v+1)\ (v+2)\ (v+3)}\right)+\frac{a_0\ z^{v+4}}{32\ (v+1)\ (v+2)}-\frac{a_0\ z^{v+2}}{4\ v+4}+a_0\ z^v$

```
/* substitution of the normalization */
besselj_artisanal1: factor(ratsubst(1/(2^v·gamma(v+1)),a[0],besselj_artisanal));
```

besselj_artisanal1 $-\left(\frac{2^{-v-7}\ z^v\ (z^6-12\ v\ z^4-36\ z^4+96\ v^2\ z^2+480\ v\ z^2+576\ z^2-384\ v^3-2304\ v^2-4224\ v-2304)}{3\ (v+1)\ (v+2)\ (v+3)\ \Gamma(v+1)}\right)$

```
besseljbydef: expand(sum((-1)^m/(gamma(m+1)·gamma(m+1+v)) · (z/2)^(2·m + v),m,0,3)) ;
```

besseljbydef $-\left(\frac{2^{-v-7}\ z^{v+6}}{3\ \Gamma(v+4)}\right)+\frac{2^{-v-5}\ z^{v+4}}{\Gamma(v+3)}-\frac{2^{-v-2}\ z^{v+2}}{\Gamma(v+2)}+\frac{z^v}{2^v\ \Gamma(v+1)}$

```
/* Without simplyfing the gammas, it's hard to see that both series expansions are equivalent.
We substitute concrete values of v to test */;
```

```
expand(ratsubst(1/2,v,besselj_artisanal1));
```

(%o27) $-\left(\frac{z^{13/2}}{315\ 2^{7/2}\ \sqrt{\pi}}\right)+\frac{z^{9/2}}{15\ 2^{5/2}\ \sqrt{\pi}}-\frac{z^{5/2}}{3\ \sqrt{2}\ \sqrt{\pi}}+\frac{\sqrt{2}\ \sqrt{z}}{\sqrt{\pi}}$

```
expand(ratsubst(1/2,v,besseljbydef));
```

(%o28) $-\left(\frac{z^{13/2}}{315\ 2^{7/2}\ \sqrt{\pi}}\right)+\frac{z^{9/2}}{15\ 2^{5/2}\ \sqrt{\pi}}-\frac{z^{5/2}}{3\ \sqrt{2}\ \sqrt{\pi}}+\frac{\sqrt{2}\ \sqrt{z}}{\sqrt{\pi}}$

```
/* The full value is */
bessel_j(1/2,z);
```

(%o29) $\frac{\sqrt{2}\ \sin(z)}{\sqrt{\pi}\ \sqrt{z}}$

```
/* which we can check for a Taylor expansion that is the same we calculated*/
taylor(bessel_j(1/2,z),z,0,8);
```

(%o30)/T/ $\frac{\sqrt{2}\ \sqrt{z}}{\sqrt{\pi}}-\frac{z^{5/2}}{3\ \sqrt{2}\ \sqrt{\pi}}+\frac{z^{9/2}}{15\ \sqrt{2}^5\ \sqrt{\pi}}-\frac{z^{13/2}}{315\ \sqrt{2}^7\ \sqrt{\pi}}+\ldots$

2 Plots

Integer index

(%i5)
```
wxplot2d([bessel_j(0,z), bessel_j(1,z), bessel_j(2,z)], [z,0,20],
 [gnuplot_postamble, "set zeroaxis;"])$
```

(%t5)



Half-integer index

(%i4) **wxplot2d([bessel_j(1/2,z), bessel_j(3/2,z), bessel_j(5/2,z)], [z,0,20],**
[gnuplot_postamble, "set zeroaxis;"])$

(%t4)



3 Example of derivation of a recurrence formula

Generating function

(%i11) `g: exp((x/2)·(t−1/t));`

g $$\%e^{\frac{\left(t-\frac{1}{t}\right)x}{2}}$$

Expansion in a Laurent series with the generating function

(%i12) `eq10: g = sum(J[n](x)·t^n,n,−inf,inf);`

eq10 $$\%e^{\frac{\left(t-\frac{1}{t}\right)x}{2}}=\sum_{n=-\infty}^{\infty}\left(t^{n}\,J_{n}(x)\right)$$

Differentiation w.r.t. t

(%i14) `eq10dt: diff(eq10,t);`

eq10dt $$\frac{\left(\frac{1}{t^{2}}+1\right)x\;\%e^{\frac{\left(t-\frac{1}{t}\right)x}{2}}}{2}=\sum_{n=-\infty}^{\infty}\left(n\,t^{n-1}\,J_{n}(x)\right)$$

We transform the indices on the r.h.s. to have the coefficients of t^n

(%i18) `newsum: changevar(rhs(eq10dt),n−1−m,m,n);`

newsum $$\sum_{m=-\infty}^{\infty} \left((m+1)\, t^m\, J_{m+1}(x)\right)$$

(%i19) newsum: changevar(newsum,m−n,n,m);

newsum $$\sum_{n=-\infty}^{\infty} \left((n+1)\, t^n\, J_{n+1}(x)\right)$$

Substitution back into the equation

(%i21) eq11: substpart(newsum,eq10dt,2);

eq11 $$\frac{\left(\frac{1}{t^2}+1\right) x\, \%e^{\frac{\left(t-\frac{1}{t}\right) x}{2}}}{2} = \sum_{n=-\infty}^{\infty} \left((n+1)\, t^n\, J_{n+1}(x)\right)$$

We substitute the generating function for its Laurent series

(%i34) eq12: 2·ratsubst(rhs(eq10),lhs(eq10),eq11);

eq12 $$\frac{(t^2+1)\, x \sum_{n=-\infty}^{\infty} \left(t^n\, J_n(x)\right)}{t^2} = 2 \sum_{n=-\infty}^{\infty} \left((n+1)\, t^n\, J_{n+1}(x)\right)$$

Now we insert the t^2 in the denominator inside of the series (a bit laborious in Maxima)

(%i35) eq13: expand(lhs(eq12));

eq13 $$\frac{x \sum_{n=-\infty}^{\infty} \left(t^n\, J_n(x)\right)}{t^2} + x \sum_{n=-\infty}^{\infty} \left(t^n\, J_n(x)\right)$$

(%i52) dpart(eq13,1,1,2);

(%o52) $$\frac{x \left(\boxed{\sum_{n=-\infty}^{\infty} \left(t^n\, J_n(x)\right)}\right)}{t^2} + x \sum_{n=-\infty}^{\infty} \left(t^n\, J_n(x)\right)$$

(%i58) sumtotransf: part(eq13,1,1);

sumtotransf $$x \sum_{n=-\infty}^{\infty} \left(t^n\, J_n(x)\right)$$

(%i61) tinside: t^(−2)·part(sumtotransf,2,1);

tinside $$t^{n-2}\, J_n(x)$$

(%i68) sumtransf: substpart(tinside,sumtotransf,2,1);

sumtransf $$x \sum_{n=-\infty}^{\infty} \left(t^{n-2}\, J_n(x)\right)$$

Now we change variables to obtain the coefficients of t^n

(%i69) sumtransf: changevar(sumtransf,n−2−m,m,n);

$$\text{sumtransf}\quad x\sum_{m=-\infty}^{\infty}\left(t^{m}\,J_{m+2}(x)\right)$$

(%i70) sumtransf: changevar(sumtransf,m−n,n,m);

$$\text{sumtransf}\quad x\sum_{n=-\infty}^{\infty}\left(t^{n}\,J_{n+2}(x)\right)$$

(%i71) eq14: substpart(sumtransf,eq13,1);

$$\text{eq14}\quad x\left(\sum_{n=-\infty}^{\infty}\left(t^{n}\,J_{n+2}(x)\right)\right)+x\sum_{n=-\infty}^{\infty}\left(t^{n}\,J_{n}(x)\right)$$

We substitute all the results into the original equation (the derivative of g(x,t) w.r.t. t)

(%i72) eq15: eq14 = rhs(eq12);

$$\text{eq15}\quad x\left(\sum_{n=-\infty}^{\infty}\left(t^{n}\,J_{n+2}(x)\right)\right)+x\sum_{n=-\infty}^{\infty}\left(t^{n}\,J_{n}(x)\right)=2\sum_{n=-\infty}^{\infty}\left((n+1)\,t^{n}\,J_{n+1}(x)\right)$$

We finally arrive at the recurrence relation

(%i76) recurrencerel1: x·J[n+2](x) + x·J[n] = 2·(n+1)·J[n+1](x);

$$\text{recurrencerel1}\quad x\,J_{n+2}(x)+J_{n}\,x=2\,(n+1)\,J_{n+1}(x)$$